pragma solidity ^0.8.0;
// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/IERC20.sol)
/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `recipient`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `sender` to `recipient` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address sender,
        address recipient,
        uint256 amount
    ) external returns (bool);

    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);
}


// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)
/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}


// OpenZeppelin Contracts v4.4.1 (access/Ownable.sol)
/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}


// OpenZeppelin Contracts v4.4.1 (utils/math/SafeMath.sol)
// CAUTION
// This version of SafeMath should only be used with Solidity 0.8 or later,
// because it relies on the compiler's built in overflow checks.
/**
 * @dev Wrappers over Solidity's arithmetic operations.
 *
 * NOTE: `SafeMath` is generally not needed starting with Solidity 0.8, since the compiler
 * now has built in overflow checking.
 */
library SafeMath {
    /**
     * @dev Returns the addition of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function tryAdd(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            uint256 c = a + b;
            if (c < a) return (false, 0);
            return (true, c);
        }
    }

    /**
     * @dev Returns the substraction of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function trySub(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            if (b > a) return (false, 0);
            return (true, a - b);
        }
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function tryMul(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            // Gas optimization: this is cheaper than requiring 'a' not being zero, but the
            // benefit is lost if 'b' is also tested.
            // See: https://github.com/OpenZeppelin/openzeppelin-contracts/pull/522
            if (a == 0) return (true, 0);
            uint256 c = a * b;
            if (c / a != b) return (false, 0);
            return (true, c);
        }
    }

    /**
     * @dev Returns the division of two unsigned integers, with a division by zero flag.
     *
     * _Available since v3.4._
     */
    function tryDiv(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            if (b == 0) return (false, 0);
            return (true, a / b);
        }
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers, with a division by zero flag.
     *
     * _Available since v3.4._
     */
    function tryMod(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            if (b == 0) return (false, 0);
            return (true, a % b);
        }
    }

    /**
     * @dev Returns the addition of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `+` operator.
     *
     * Requirements:
     *
     * - Addition cannot overflow.
     */
    function add(uint256 a, uint256 b) internal pure returns (uint256) {
        return a + b;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b) internal pure returns (uint256) {
        return a - b;
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `*` operator.
     *
     * Requirements:
     *
     * - Multiplication cannot overflow.
     */
    function mul(uint256 a, uint256 b) internal pure returns (uint256) {
        return a * b;
    }

    /**
     * @dev Returns the integer division of two unsigned integers, reverting on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator.
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b) internal pure returns (uint256) {
        return a / b;
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * reverting when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b) internal pure returns (uint256) {
        return a % b;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting with custom message on
     * overflow (when the result is negative).
     *
     * CAUTION: This function is deprecated because it requires allocating memory for the error
     * message unnecessarily. For custom revert reasons use {trySub}.
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(
        uint256 a,
        uint256 b,
        string memory errorMessage
    ) internal pure returns (uint256) {
        unchecked {
            require(b <= a, errorMessage);
            return a - b;
        }
    }

    /**
     * @dev Returns the integer division of two unsigned integers, reverting with custom message on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(
        uint256 a,
        uint256 b,
        string memory errorMessage
    ) internal pure returns (uint256) {
        unchecked {
            require(b > 0, errorMessage);
            return a / b;
        }
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * reverting with custom message when dividing by zero.
     *
     * CAUTION: This function is deprecated because it requires allocating memory for the error
     * message unnecessarily. For custom revert reasons use {tryMod}.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(
        uint256 a,
        uint256 b,
        string memory errorMessage
    ) internal pure returns (uint256) {
        unchecked {
            require(b > 0, errorMessage);
            return a % b;
        }
    }
}


// OpenZeppelin Contracts v4.4.1 (security/ReentrancyGuard.sol)
/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        // On the first call to nonReentrant, _notEntered will be true
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;

        _;

        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }
}

abstract contract CharityAcceptor {
    mapping(address => address[]) public getAcceptors;

    function getAcceptorsLength(address token) public view returns (uint) {
        return getAcceptors[token].length;
    }

    function _setCharityAcceptors(
        address token,
        address[] calldata wallets
    ) internal returns (uint) {
        for (uint i = 0; i < wallets.length; i++) {
            if (_isValidAddress(wallets[i])) {
                (bool found, ) = _indexOf(getAcceptors[token], wallets[i]);
                if (!found) {
                    getAcceptors[token].push(wallets[i]);
                }
            }
        }
        return getAcceptors[token].length;
    }

    function _addCharityAcceptor(
        address token,
        address wallet
    ) internal returns (bool) {
        if (_isValidAddress(wallet)) {
            (bool found, , ) = _indexOf(token, wallet);
            if (!found) {
                getAcceptors[token].push(wallet);
                return true;
            }
        }
        return false;
    }

    function _removeCharityAcceptor(
        address token,
        address wallet
    ) internal returns (bool, uint) {
        (bool found, uint index, address[] memory _wallets) = _indexOf(
            token,
            wallet
        );
        if (found) {
            for (uint i = index; i < _wallets.length - 1; i++) {
                getAcceptors[token][i] = _wallets[i + 1];
            }
            getAcceptors[token].pop();
            return (true, index);
        }
        return (false, 0);
    }

    function _indexOf(
        address token,
        address wallet
    ) internal view returns (bool, uint, address[] memory) {
        address[] memory _wallets = getAcceptors[token];
        (bool found, uint index) = _indexOf(_wallets, wallet);
        return (found, index, _wallets);
    }

    function _indexOf(
        address[] memory _wallets,
        address wallet
    ) internal pure returns (bool, uint) {
        for (uint index = 0; index < _wallets.length; index++) {
            if (_wallets[index] == wallet) return (true, index);
        }
        return (false, 0);
    }

    function _isValidAddress(address wallet) internal view returns (bool) {
        return wallet != address(0) && wallet != address(this);
    }

    modifier hasAcceptors(address token) {
        require(
            getAcceptors[token].length > 0,
            "CharityWallet: Invalid acceptors"
        );
        _;
    }
}

abstract contract DistributionAmount {
    mapping(address => uint256) public getMinAmount;

    function _setMinAmount(
        address token,
        uint256 amount
    ) internal returns (bool) {
        if (_isValidAmount(amount)) {
            getMinAmount[token] = amount;
            return true;
        }
        return false;
    }

    function _isValidAmount(uint256 amount) internal pure returns (bool) {
        return amount > 0;
    }

    modifier hasMinAmount(address token) {
        require(
            _isValidAmount(getMinAmount[token]),
            "CharityWallet: Invalid amount"
        );
        _;
    }
    
}

abstract contract CharityDistributor {

    event FundsTransfered(
        address indexed token,
        address indexed acceptor,
        uint256 amount,
        uint256 totalAmount
    );

    using SafeMath for uint256;

    mapping(address => mapping(address => uint256)) public getAmountSent;
    mapping(address => uint256) internal _lastIndex;

    function _balanceOf(address token) internal view returns (uint256) {
        return IERC20(token).balanceOf(address(this));
    }

    function _transfer(
        address token,
        address[] memory wallets,
        uint256 amount
    ) internal {
        // trasfer amount token to wallets
        for (uint i = 0; i < wallets.length; i++) {
            _transfer(token, wallets[i], amount);
        }
    }

    function _transfer(address token, address wallet, uint256 amount) internal {
        // trasfer amount token to single wallet
        if (IERC20(token).transfer(wallet, amount)) {
            uint256 totalAmount = getAmountSent[token][wallet].add(amount);
            getAmountSent[token][wallet] = totalAmount;
            emit FundsTransfered(token, wallet, amount, totalAmount);
        }
    }

    function _distribute(
        address token,
        address[] memory wallets,
        uint256 minAmount
    ) internal {
        uint256 balance = _balanceOf(token);
        // distrubute funds only if sufficient balance has been accumulated
        if (balance >= minAmount) {
            if (balance >= minAmount.mul(wallets.length)) {
                // distrubute same amount to all wallets. _lastIndex[token] remains unchanged
                uint256 amount = balance.div(wallets.length);
                // transfer funds
                _transfer(token, wallets, amount);
            } else {
                // distribute minAmount to wallets untill balance is sufficient
               
                while (balance >= minAmount) {
                     uint index = _lastIndex[token];
                    _transfer(token, wallets[index], minAmount);
                    // substract minAmount from balance
                    balance = balance.sub(minAmount);
                    // update last index
                    _lastIndex[token] = ++index < wallets.length ? index: 0;
                  
                }
            }
        }
    }
}

abstract contract CharityToken {
    address[] public getToken;

    function getTokenCount() public view returns (uint) {
        return getToken.length;
    }

    function _addToken(address token) internal returns (bool) {
        if (_isValidToken(token)) {
            (bool found, ) = _findToken(token);
            if (!found) {
                getToken.push(token);
                return true;
            }
        }
        return false;
    }

    function _removeToken(address token) internal returns (bool) {
        (bool found, uint index) = _findToken(token);
        if (found) {
            for (uint i = index; i < getToken.length - 1; i++) {
                // shift
                getToken[i] = getToken[i + 1];
            }
            // remove last element
            getToken.pop();
            return true;
        }
        return false;
    }

    function _isValidToken(address token) internal view returns (bool) {
        return token != address(0) && token != address(this);
    }

    function _findToken(address token) internal view returns (bool, uint) {
        for (uint i = 0; i < getToken.length; i++) {
            if (getToken[i] == token) return (true, i);
        }
        return (false, 0);
    }

    modifier tokenExists(address token) {
        (bool found, ) = _findToken(token);
        require(found, "CharityWallet: Invalid token");
        _;
    }
}

contract CharityWallet is
    Ownable,
    ReentrancyGuard,
    CharityAcceptor,
    CharityDistributor,
    CharityToken,
    DistributionAmount
{
    // prevent ETH to be sent to contract
    fallback() external payable {
        revert("CharityWallet: Forbidden");
    }

    receive() external payable {
        revert("CharityWallet: Forbidden");
    }

    /**
     * @dev add `token` to distibution
     *
     * Requirements:
     *
     * - caller must be contract onwer
     * - non reentrant
     */

    function addToken(
        address token,
        uint256 minAmount,
        address[] calldata wallets
    ) external onlyOwner nonReentrant {
        require(_addToken(token), "CharityWallet: Invalid token");
        require(
            _setCharityAcceptors(token, wallets) > 0,
            "CharityWallet: Invalid acceptors"
        );
        require(
            _setMinAmount(token, minAmount),
            "CharityWallet: Invalid amount"
        );
    }

    /**
     * @dev remove `token` from distibution
     *
     * Requirements:
     *
     * - caller must be contract onwer
     * - valid token
     * - non reentrant
     */

    function removeToken(
        address token
    ) external onlyOwner tokenExists(token) nonReentrant {
        _removeToken(token);
        delete getAcceptors[token];
        delete getMinAmount[token];
    }

    /**
     * @dev set minimal `amount` to be distributed for `token`
     *
     * Requirements:
     *
     * - caller must be contract onwer
     * - valid token
     * - non reentrant
     */

    function setMinAmount(
        address token,
        uint256 amount
    ) external onlyOwner tokenExists(token) nonReentrant {
        require(_setMinAmount(token, amount), "CharityWallet: Invalid amount");
    }

    /**
     * @dev add charity acceptor `wallet` for `token`
     *
     * Requirements:
     *
     * - caller must be contract onwer
     * - valid token
     * - non reentrant
     */

    function addCharityAcceptor(
        address token,
        address wallet
    ) external onlyOwner tokenExists(token) nonReentrant {
        require(
            _addCharityAcceptor(token, wallet),
            "CharityWallet: Invalid acceptor"
        );
    }

    /**
     * @dev remove charity acceptor `wallet` for `token`
     *
     * Requirements:
     *
     * - caller must be contract onwer
     * - valid token
     * - non reentrant
     */

    function removeCharityAcceptor(
        address token,
        address wallet
    ) external onlyOwner tokenExists(token) nonReentrant {
        (bool success, uint index) = _removeCharityAcceptor(token, wallet);
        if (success) {
            if (index > 0 && index >= _lastIndex[token]) {
                _lastIndex[token] = index - 1;
            }
        } else {
            revert("CharityWallet: Invalid acceptor");
        }
    }

    /**
     * @dev can distribute `token` to acceptors
     */

    function canDistributeToken(address token) public view returns (bool) {
        (bool found, ) = _findToken(token);
        if (found) {
            uint256 minAmount = getMinAmount[token];
            if (_isValidAmount(minAmount)) {
                return _balanceOf(token) >= minAmount;
            }
        }
        return false;
    }

    /**
     * @dev can distribute at least one token
     */

    function canDistribute() public view returns (bool) {
        for (uint i = 0; i < getToken.length; i++) {
            if (canDistributeToken(getToken[i])) return true;
        }
        return false;
    }

    /**
     * @dev distribute `token` to acceptors
     *
     * Requirements:
     *
     * - valid token
     * - valid min amount
     * - has at least one acceptor
     * - non reentrant
     */

    function distributeToken(
        address token
    )
        external
        tokenExists(token)
        hasMinAmount(token)
        hasAcceptors(token)
        nonReentrant
    {
        CharityDistributor._distribute(
            token,
            getAcceptors[token],
            getMinAmount[token]
        );
    }

    /**
     * @dev distribute accumulated tokens.
     */

    function distribute() external  {
        for (uint i = 0; i < getToken.length; i++) {
            address token = getToken[i];
            if (canDistributeToken(token)) {
                _distribute(token, getAcceptors[token], getMinAmount[token]);
            }
        }
    }
}